
Part 1 of  Building building “stuff” in POVRay 

Getting Started.
OK - there’s some initial set-up stuff you’ll need to do
depending on where you begin!
1) if you don’t have it, download the latest version of
POVray from their site!
2) set up a working directory somewhere?
3) Run POVray
4) select “Render - Edit Settings/Render” and set the
Source File location to your working directory then hit
“Set but don’t Render”
[minor thing – Under Editor-Preferences set the Tab
setting to something like 4]

 

Set Up Files.
Hit New (if there isn’t a blank document already open)
Copy the text on the right here into your new
document.
Save As - “tutorial.pov”

#include "colors.inc"
#include "gridsetc.inc"
 
 
// switches and stuff
#declare axes = 0.2;
#declare grid = 0.1
#declare noshadow = 1;;
#declare rd=seed(31574);

 
// camera location
camera {
 location  < 100.0, 50.0, -200.0>
 look_at   <0.0, 10.0,  0.0>
 angle      35
}
 
// illumination
light_source { < 200,400,-200> White}
light_source { <-200, 50,-400> White}
 
background {SkyBlue}
 
plane { y, 0.0 pigment {ForestGreen}}
 
#if (axes > 0) object{axes1(axes,noshadow) translate < 0, 0.0, 0> }
#end
#if (grid > 0) object{gridsp(grid) translate < 0, 0.0, 0> } #end  
 
// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ##########

 Hit new (if there isn’t a blank document already open)
Copy the text on the right here into your new
document.
Save As – "gridsetc.inc"

Just trust me, it may look complicated but worry not.
These are merely things to help the tutorial.

You should now have two POV documents.
“tutorial.pov” - where we will build our stuff
and
“gridsetc.inc” - an “include” file containing some helpful
declarations and macros.

A macro is just a sort of home-made command. More
on that later!

#declare tr       = 0.65;
#declare Redt     = rgbt <1, 0, 0, tr>;
#declare Greent   = rgbt <0, 1, 0, tr>;
#declare Bluet    = rgbt <0, 0, 1, tr>;
#declare Yellowt  = rgbt <1, 1, 0, tr>;
#declare Cyant    = rgbt <0, 1, 1, tr>;
#declare Magentat = rgbt <1, 0, 1, tr>;
#declare Whitet   = rgbt <1, 1, 1, tr>;
#declare Blackt   = rgbt <0, 0, 0, tr>;

#macro axes1(thick,noshadow)
union {
    #declare thick= axes;
    //positive axes  
    cylinder { < 1200,  0,  0>, <- 0,  0,  0> thick pigment {Red} }
// +x-axis
    cylinder { <  0, 1200,  0>, <  0,  0,  0> thick pigment {Green} }
// +y-axis
    cylinder { <  0,  0, 1200>, <  0,  0,  0> thick pigment {Blue} }
// +z-axis
    //positive axes                                                   
    cylinder { <  0,  0,  0>, <-1200,  0,  0> thick pigment {SpicyPink}
}   // -x-axis
    cylinder { <  0,  0,  0>, <  0,-1200,  0> thick pigment
{YellowGreen} } // -y-axis
    cylinder { <  0,  0,  0>, <  0,  0,-1200> thick pigment {Cyan} }
// -z-axis
    #if (noshadow = 1) no_shadow #end
} 
#end  

#macro gridsp(thick) 
union{
    #declare xs = -100;
    #while (xs <= 100)
        cylinder { < 100,  0,   xs>, <-100,  0,    xs> thick pigment
{Orange} }
        cylinder { <  xs,  0,  100>, <  xs,  0,  -100> thick pigment
{Orange} }
        #declare xs = xs +10;
    #end
}
#end 

 So lets render something!
Select the “tutorial.pov” as your current file using the
tabs
Just above the file tabs, on the left, is a drop box for
choosing various image sizes. This tells POVRay what
size of image to render into.
Select something around 800x600, No AA 
[AA = Anti Aliasing – more later maybe]

 



You should get this.... which is kinda nice.
so if we look at our pov file what does it say?

 

#include "colors.inc"
#include "gridsetc.inc" The #include command tells POV to use declarations it

can find in these additional files.
“colors.inc” comes with POVRay and defines a whole load
of RGB colours for us that we can call up by name.
“gridsetc.inc” is the file we just created and contains some
custom colours and the axes and grid macros.

// switches and stuff
#declare axes = 0.2;
#declare grid = 0.1;
#declare noshadow = 1;
#declare rd=seed(31574);

 

[Firstly, “//” is a comment marker, anything on the same
line after this is ignored by POV]
#declare does what it says on the tin, you have to use this
to assign a value to a variable.
The axes and grid variables are my own ones used  to
define the thickness of the lines used on the axes and
grid. By having a non-zero value they also switch ON the
respective thing.
rd is a variable we can use when calling up the random
function rand( ) and this particular command plants a seed
for generating those numbers – it's a default sort of thing
to do. The seed can any number you like (not sure what
the maximum is, mind!)

// camera location
camera {
 location  < 100.0, 50.0, -200.0>
 look_at   <0.0, 10.0,  0.0>
 angle      35
}
 
/

 The camera{    } command calls up our viewpoint using a
location and a look_at point. I'll explain the coordinates
system in just a sec.
The “angle” parameter sets the view angle of the camera
in degrees (horizontal angle I think). 

/ illumination
light_source { < 200,400,-200> White}
light_source { <-200, 50,-400> White}
 

We have camera, now lights!
Very simple, a position and a colour. As many as you like,
I just have the two for now. These are simple point lights,
they cannot themselves be seen, however.

background {SkyBlue}
 
plane { y, 0.0 pigment {ForestGreen}}
 

Some scenery (ehem).
Background sets the, well, background colour. Anywhere
that isn't something basically.

The “plane”, now here's our first visible “thing”. A simple
flat surface. The “y” tells it to be normal (at right angles in
all directions) to the y-axis. In this scene, y is “up”. The
“0.0” tells it to be at the y=0.0 position. 
Pigment tells it what colour to be, ForestGreen is one of
the colours defined in “colors.inc”, like White earlier.

#if (axes > 0) object{axes1(axes,noshadow) translate < 0, 0.0, 0> }
#end
#if (grid > 0) object{gridsp(grid) translate < 0, 0.0, 0> } #end  

 And these are my own bit of helpfulness.
It might be a tad pre-emptive here to go into details.
Basically, depending on the switches set further up they
put the axes and/or the grid on the scene as guidelines. In
addition, you can set the axes to cast no shadows.

// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ##########

 A comment block.



 So, what does it show?
Have a look at the axes.
I have colour coded these and this camera sort of view
angle is more or less the one I generally tend to use.
Simply for this reason.
If you look at the Green and Red lines these are +y
and +x so this view puts them in the traditional
directions one might see for a graph or what teacher
drew on the blackboard. 
Everything else follows from there and sadly it means
that +z is “into the screen” or away from us (Dark Blue)
By personal convention here I have made the -ve
versions of each axis a lighter shade of the colour of
the +ve version. 

The Grid has a line spacing of 10 POV Units.

 

 Ah, units. A variable beast, as it were. Some people
do ALL their designs inside a unit box, others, like me,
are less organised in their thinking and define what
POVRay Units represent depending on what's easiest
for the job.
For example, I did this train. The drawing I worked from
was in feet and inches, so I did the design by relating 1
pov unit = 1 inch.
It's all much of a muchness really, firstly, you can
rescale objects when you call them up and secondly
units are floating point numbers and have a huge
degree of accuracy, if  needed.... within reason

In this tutorial we shall use 1 POV Unit = 1 meter. Give
or take.

 

 
Enough of all this! Let's do something.

 

 We're going to build a very simplistic Keep. So let's
start with a tower, and go for a simple one at that.

 I warn you in advance – I'm going to be rather half-assed
in the way I do things simply because that way I get to be
able to show you more of the features of POVRay. I would
not necessarily recommend writing your own stuff like
some of this, and if you do, don't show it to your friends!!!

 POVRay can render stuff like meshes, if it wants to, it
isn't really designed to create/edit them so most of the
time, a simple soul like me muddles through with its set
of primitive shapes. That train earlier, it has 3 meshes
buried in there for rather more complex bits – the
chimneys, the firebox and a small bracket right at the
front. All the rest is primitives.

 Standard basic POV primitive shapes that I get away with most of the
time are : 
sphere, box, cylinder, cone and torus (donut!). 

Oh, and the plane we've already seen, but that's the only place I use
that.
There are a whole bunch of other shapes I use more rarely, that's just
me.

 It's a square tower, so we'll use a box here.  

There's a handy little tool here, so get the edit cursor
after the PUT STUFF HERE comment and select
“Insert - Shapes - Box”
It more or less explains the thing here in comments.

box {
  <-1, -1, -1>  // one corner position <X1 Y1 Z1>
  < 1,  1,  1>  // other corner position <X2 Y2 Z2>
}

Generally this Insert drop-down tool gives you a sort of
basic “unit” sized version of the thing, un-pigmented. If
you use “Insert – Shapes (shorter versions)” you get
much the same thing but without all the comments and
packed to one line as appropriate.

 



Anyhoo – a box is defined by telling POV two of its
opposing corners. Doesn't matter much which, but I
tend to use most negative and positive, in that order.
Usually....

Anyway, the coordinate convention is < x, y, z>, so it's
pretty simple. 
Our “tower” will be (say) 6 meters square and (what?)
12 meters tall and sitting on the plane we've got? Let's
make it Yellow.

With this
// create a box that extends between the 2 specified points
box {
  <-3,  0, -3>  // one corner position <X1 Y1 Z1>
  < 3, 12,  3>  // other corner position <X2 Y2 Z2>  
  pigment {Yellow}
}

If you're like me and hate excessive comments kicking
around, the shorter version would look like this.

box { <-3,  0, -3> < 3, 12,  3> pigment {Yellow} }

Hrm.
It's a bit small looking so let's adjust our camera.

 We can make this thing look bigger two ways.
1) move the camera closer (doh!)
2) reduce the angle of view (zoom)

So, to move closer we can just divide our camera's
position by (say) 4. The position <x, y, z> is a Vector –
you can treat it like most “ordinary” numbers in many
ways so 
// camera location
camera {
  location  < 100.0, 50.0, -200.0>/4
  look_at   <0.0, 10.0,  0.0>
  angle      35
}

is perfectly valid, if a tad lazy? 

 Just to belabour the point, this view is a bit low down
as we can't see the top of the box so we could do this
// camera location
camera {
  location  < 100.0, 50.0, -200.0>/4 + 10*y
  look_at   <0.0, 10.0,  0.0>
  angle      35
}

The “*y” means “in the y-component” of the vector, the
same as +< 0, 10, 0>. It's all pretty flexible.

However, I have a sense of “appropriateness”-ness
and I don't particularly like doing these sorts of things –
except temporarily... so I've set mine to the same result
in the original vector.
// camera location
camera {
  location  < 25.0, 22.5, -50.0>
  look_at   <0.0, 10.0,  0.0>
  angle      35
}

 OK, this is fine if towers are just solid blocks, but...
well, most aren't...

So let's hollow it out to be 4 walls. We can do that
using the snappily named concept of “Constructive
Solid Geometry” or CSG.  These allow us to take
multiple objects and “do stuff” together with them,
mainly union, intersection, difference and merge. We're
gonna use  Difference.

 



 Make another box, smaller in x and z by the thickness
of our “walls” (1m here) and a bit bigger in the z
directions (any old number really). Make it Blue and
add it after our yellow one.
Like this:

box { <-3,   0, -3> < 3, 12,  3> pigment {Yellow} }
box { <-2,  -1, -2> < 2, 13,  2> pigment {Blue} }

as one might expect, you just see part of it.

Now for a slightly clever bit

 In our special INC file I've declared some rather
transparent versions of these colours. 
Add a “t” after each colour so they read “Yellowt” and
“Bluet”

Cool, huh?

Now let's rip with the CSG carving knife.
Surround these two box statements with a 
difference {..} command
thus
difference{
box { <-3,   0, -3> < 3, 12,  3> pigment {Yellowt} }
box { <-2,  -1, -2> < 2, 13,  2> pigment {Bluet} }
}

What this command has done is take the first shape in
the list between the {..}'s and cut away all bits that
coincide with any of the subsequent shapes.
I can add more shapes here and it doesn't much
matter what order I do them in – only that the one “cut
from” is first in the list.

Another important aspect of this command is that the
difference{…} combination – or CSG – can now be
treated as a single object. You could use this CSG to
cut away stuff from another object  with all the
concomitant complications of perceptual “double
negatives” and so on that implies...

 We can also pigment it as a single entity.

difference{
    box { <-3,   0, -3> < 3, 12,  3> }
    box { <-2,  -1, -2> < 2, 13,  2> }
    pigment {Yellowt}
}

Since it is now essentially a single “thing” made from 2
others.
You will also note that I have indented stuff after the “{“
to keep things, well, tidy. 



Let's move on. Let's do some arrow slits.
Firstly, we don't need our current tower cluttering stuff
up, so we comment it out.
The way to comment out a block of text is by using 
/*
..
..
*/

and everything in between these two markers is
ignored.

 I'm going to want to see some negative y stuff so I'm
commenting out our plane

//plane { y, 0.0 pigment{ForestGreen} }

AND!!! Since the thing I want to play with is smaller and
centred around the origin I'm changing the camera's “look-
at” point to < 0, 0, 0> (from < 0, 10, 0>)

 So for an arrow slit we want a sort of wedge shape we
can “difference” with our current tower to make holes of
the right sort of shape. Let's start with a box

box { <-1, -2, -4>, <1, 4, 0> pigment{Red} }

Let's look at this a moment, I have positioned this
thinking ahead. 
One side is flush up against the x and y axes. In this
case by having that 0 in there in the z-coordinate –
we'll see why in a mo.

 Now I create another box, bigger than the red one in
all directions but only overlapping half of it, kinda
offset.

box { <-1, -3, -5>, <0, 5, 1> pigment{Greent} }

We're going to use this for differencing stuff.
But first – rotation!

 When you rotate stuff, it always happens around an
axis. Always.
This means that if you want to have things floating out
there rotated, you build the thing at the axis, rotate it,
then move it. UNLESS, you don't. 
All may become clearer.
So

 

Let's rotate our Greent thing 10 degrees around the y-
axis (the up/down axis)

box { <-1, -3, -5>, <0, 5, 1> rotate 10*y pigment{Greent} }

I usually get the direction wrong! This time I was right.



If we now difference this from the red box

difference{
    box { <-1, -2, -4>, <1, 4, 0> pigment{Red} }
    box { <-1, -3, -5>, <0, 5, 1> rotate 10*y pigment{Greent} }
}

We can see that we've hacked a diagonal from the far
side.

Replicating the Greent box on the other side gives us

difference{
    box { <-1, -2, -4>, <1, 4, 0> pigment{Red} }
    box { <-1, -3, -5>, <0, 5, 1> rotate 10*y pigment{Greent} }
    box { < 1, -3, -5>, <0, 5, 1> rotate -10*y pigment{Greent} }
}

Interesting!! What you see here is one of the properties
of difference that we can exploit sometimes. The facet
you create when chopping lumps off like this derives its
colour from the differenced object. Can be useful.

Regardless – we have the basis of our wedgie shape.

So, this is kind of like an arrow slit for horizontal use,
but we need the down angle too. This is why I chose
this red block to extend into the negative y.

I create a third block down there.
//difference{
    box { <-1, -2, -4>, <1, 4, 0> pigment{Red} }
    
    box { <-1, -3, -5>, <0, 5, 1> rotate 10*y pigment{Greent} }
    box { < 1, -3, -5>, <0, 5, 1> rotate -10*y pigment{Greent} }

    box { <-2,  -3, -5>, <2, 0, 1> pigment{Yellow} }
//}

Let's look at this carefully. I made the block
methodically, looking at each of the three coordinate
pairs in turn. Our red block goes from +1 to -1 in the x
direction, so I wanted bigger than that, so 2 and -2.
The -5 to 1 in z was the same as the other Greent
blocks, so keep those.
In the y direction, I only want to slice off from 0
downwards, and I need to go further down than the
red's -2 so I chose -3. Good enough to start with.

So, let's rotate the yellow block now. This time around
the x-axis (red/pink)
//difference{
    box { <-1, -2, -4>, <1, 4, 0> pigment{Red} }
    
    box { <-1, -3, -5>, <0, 5, 1> rotate 10*y pigment{Greent} }
    box { < 1, -3, -5>, <0, 5, 1> rotate -10*y pigment{Greent} }

    box { <-2,  -3, -5>, <2, 0, 1> rotate -25*x pigment{Yellow} }
//}

Looks about right, let's put the difference back and lose
those transparent colours
.



difference{
    box { <-1, -2, -4>, <1, 4, 0> pigment{Red} }
    
    box { <-1, -3, -5>, <0, 5, 1> rotate 10*y pigment{Green} }
    box { < 1, -3, -5>, <0, 5, 1> rotate -10*y pigment{Green} }

    box { <-2,  -3, -5>, <2, 0, 1> rotate -25*x pigment{Yellow} }
}

We've made another CSG. Coo.
We can make this a multi-use object too by giving it a
name.

#declare wedgie =  difference{
    box { <-1, -2, -4>, <1, 4, 0> pigment{Red} }
    
    box { <-1, -3, -5>, <0, 5, 1> rotate 10*y pigment{Green} }
    box { < 1, -3, -5>, <0, 5, 1> rotate -10*y pigment{Green} }

    box { <-2,  -3, -5>, <2, 0, 1> rotate -25*x pigment{Yellow} }
}

If we ran this, it would disappear, as this code block is
now a declaration, not an “instantiation”. We now need
to call it up.

To do this we write

object {wedgie}

Let's talk about this a second. I said “give it a name” but in reality we used the same #declare command we used
to set a value to a variable. This is, in fact, the same thing. We have given the variable “wedgie” in this case, a
“value” of a CSG. In all operations where CSGs can be used with each other, we can use this variable name
“wedgie” to represent this shape.
POVray is like a number of languages, very flexible about variable “types”. You don't have to declare what type a
variable is (except with arrays) and it's clever enough to work out how to use ostensibly different variable and
constant types together. Like in our earlier example of dividing a vector “type” by 4 by using “< x, y, z>/4”.
I won't go into this further here, let's just say that if POVray get's upset about something in this area, it'll let you
know AND if you want to get adventurous, be sure to look out for it doing “the right thing” when you might be
hoping it wouldn't!

Back to the build.
We have a wedge shape and a simple tower. Let's see
them together and uncomment the tower.

As a matter of good practice, I move the wedgie
declaration infront of the tower code. Always declare
stuff first (or in an include file) and have instance calls
after – POV doesn't read minds, or “look ahead” for
things.

This is the code I have at the moment:

#declare wedgie = difference{
    box { <-1, -2, -4>, <1, 4, 0> pigment{Red} }
    
    box { <-1, -3, -5>, <0, 5, 1> rotate 10*y pigment{Green} }
    box { < 1, -3, -5>, <0, 5, 1> rotate -10*y pigment{Green} }

    box { <-2,  -3, -5>, <2, 0, 1> rotate -25*x pigment{Yellow} }
}    
    

difference{
    box { <-3,   0, -3> < 3, 12,  3>  }
    box { <-2,  -1, -2> < 2, 13,  2>  }
    
    pigment {Yellowt}
}

object{wedgie}

Clearly my sense of proportions is insane!
Actually, it is, but that's private. Sometimes reality is
bigger or smaller than you imagine. 
My tower is a bit modest, but not outragously tiny.
However, my arrowslit is rather more suited to a
cannon. I could go back and jigger around with all
those wretched coordinates – OR I could just scale it.

Scaling. I'm guessing purists would cast one into the pit of
2D-mons for even thinking such thoughts, but it has its
uses and this is a way of showing how to do it.

Feeble excuse! Banish him to DOS shell!



Scale
It's quite easy, and sometimes you can get interesting
and unexpected results – especially with more complex
shapes than this one!

Try 

object{wedgie scale 0.5}

this is a cheaty sort of way of saying <0.5, 0.5, 0.5> =
scale in all directions.

You should be able to see here that the thing has
reduced in size around the origin. This is important. If
our object were offset from the origin by some
distance, then that offset would also be scaled. 
Another reason for building things at the origin and
scale, rotate THEN move. 

Everything looks a bit small to work on here. Let's
reconfigure a few things in our view

Let's “look” up a bit. Our tower is 12m tall, so change
the “look_at” of our camera to <0.0, 6.0,  0.0> (half way
up)

let's use zoom this time by changing the camera
“angle” to 20.
Finally, to reduce our grid and axes thickness, change
the declarations of “axes” and “grid” to 0.1 and 0.05
respectively.

Better?

OK – moving stuff.
Otherwise known as “translate”

Let's move our wedgie a bit. (I've also recoloured all its
cut-outs as Blue)

object {wedgie scale 0.5 translate < 0, 11, -2>}

I've lifted it right up to the top so I can see how much
our wedge intrudes through our wall with the -2 z-
direction move. Hrm, A bit far, perhaps back a bit,
0.5m.
(If you remember the maths, you'd have got this
without trial and error. The wedge's sharp edge was
built along the y-axis and our inner wall is 2m from
that)

Anyway, I'm not happy with the scale entirely either,
the wedge is too tall, let's change scale 0.5 to < 0.5,
0.4, 0.5>.

object {wedgie scale < 0.5, 0.4, 0.5> translate < 0, 11, -1.5>}

Yeah.



OK, let's put this somewhere more sensible and move
it inside the tower's difference block. I feel a sudden
need to add a comment too !

// TOWER
difference{
    box { <-3,   0, -3> < 3, 12,  3>  }
    box { <-2,  -1, -2> < 2, 13,  2>  }
    
    object {wedgie scale < 0.5, 0.4, 0.5> translate < 0, 8, -1.5>}

    pigment {Yellowt}
}

At this point, all our colours are detracting from our
view of this thing. What I usually do at this point is
comment out a lot of the individual pigment statements
– by ending the object call with a “} //pigment....” etc  - I
can bring them back later if I want to. As both these are
CSGs I can give them overall colour, and Grey is kinda
traditional for castles?
And to get our “ground” back I've uncommented the
plane.

My code is now

// ########## PUT STUFF HERE ########## 

// ARROW SLIT CUTOUT
#declare wedgie = difference{
    box { <-1, -2, -4>, <1, 4, 0> }// pigment{Red} }    
    box { <-1, -3, -5>, <0, 5, 1> rotate 10*y }// pigment{Blue} }
    box { < 1, -3, -5>, <0, 5, 1> rotate -10*y }// pigment{Blue} }

    box { <-2,  -3, -5>, <2, 0, 1> rotate -25*x }// pigment{Blue} }
    pigment {Grey}
}
    
    
// TOWER
difference{
    box { <-3,   0, -3> < 3, 12,  3>  }
    box { <-2,  -1, -2> < 2, 13,  2>  }
    
    object {wedgie scale < 0.5, 0.4, 0.5> translate < 0, 8, -1.5>}

    pigment {Grey}
}

That's all for Part 1: In Part 2 I'll go into how to code in more arrow slits, rather than just copy the object manually
and how to put textures and “normal” effects in there too. Maybe some other things!
Oliver Harlow  14 Mar 2011

Disclaimer: I am not a software engineer by a long shot – I have written stuff in VHDL – a hardware language and
some of my terminology stems from there. I still feel a need to wash my mouth out after using some of them! 
Anyway, feel free to correct me on my terms usage, and I will reciprocate by probably ignoring it! Just so long as
you get the gist of what I'm saying, that's fine.
I am also no expert on POVray, it's a very capable tool and I have not explored all it's secrets yet.


